

Parse Tree

PROGRAM

```
<p4program> := <declarationList>

<declarationList> := { <declaration> | ; /* empty declaration */ }*

<declaration> := <variableDeclaration> | <externDeclaration> | <actionDeclaration> |
    <typeDeclaration> | <parserDeclaration> | <controlDeclaration> | <instantiation> |
    <errorDeclaration> | <matchKindDeclaration> | <functionDeclaration>

<nonTypeName> := IDENTIFIER | apply | key | actions | state | entries | type

<name> := <nonTypeName> | TYPE_IDENTIFIER

<nonTableKwName> := IDENTIFIER | TYPE_IDENTIFIER | apply | state | type

<parameterList> := { <parameter> { , <parameter> }* }

<parameter> := { <direction> } <typeRef> <name> { = <expression> }

<direction> := in | out | inout

<packageTypeDeclaration> := package <name> <optTypeParameters> ( <parameterList> )

<instantiation> := <typeRef> ( <argumentList> ) <name> ;

<optConstructorParameters> := { ( <parameterList> ) }
```

PARSER

```
<parserDeclaration> := <parserTypeDeclaration> <optConstructorParameters>
    { <parserLocalElements> <parserStates> }

<parserLocalElements> := { <parserLocalElement> }*

<parserLocalElement> := <variableDeclaration> | <instantiation> /* | <valueSetDeclaration> */

<parserTypeDeclaration> := parser <name> <optTypeParameters> ( <parameterList> )

<parserStates> := <parserState> { <parserState> }*

<parserState> := state <name> { <parserStatements> <transitionStatement> }

<parserStatements> := { <parserStatement> }*

<parserStatement> := <assignmentOrMethodCallStatement> | <directApplication> | <parserBlockStatement> |
    <variableDeclaration> | <emptyStatement>

<parserBlockStatement> := { <parserStatements> }

<transitionStatement> := { transition <stateExpression> }

<stateExpression> := <name> ; | <selectExpression>

<selectExpression> := select ( <expressionList> ) { <selectCaseList> }

<selectCaseList> := { <selectCase> }*

<selectCase> := <keysetExpression> : <name> ;
```

```

<keysetExpression> := <tupleKeysetExpression> | <simpleKeysetExpression>

<tupleKeysetExpression> := ( <keysetExpressionList> )

<keysetExpressionList> := <simpleKeysetExpression> { , <simpleKeysetExpression> }*

<simpleKeysetExpression> := <expression> /* { ( mask | range ) <expression> } */ | default | _

/* <valueSetDeclaration> := valueset '<' ( <baseType> | <tupleType> | <typeName> ) '>'
   '(' <expression> ')' <name> ';' */

```

CONTROL

```

<controlDeclaration> := <controlTypeDeclaration> <optConstructorParameters>
    { <controlLocalDeclarations> apply <blockStatement> }

<controlTypeDeclaration> := control <name> <optTypeParameters> ( <parameterList> )

<controlLocalDeclaration> := <variableDeclaration> | <actionDeclaration> | <tableDeclaration> |
    <instantiation>

<controlLocalDeclarations> := { <controlLocalDeclaration> }*

```

EXTERN

```

<externDeclaration> := extern
    ( <nonTypeName> <optTypeParameters> { <methodPrototypes> } ) |
    ( <functionPrototype> ; )

<methodPrototypes> := { <methodPrototype> }*

<functionPrototype> := <typeOrVoid> <name> <optTypeParameters> ( <parameterList> )

<methodPrototype> :=
    ( <functionPrototype> ; ) |
    ( TYPE ( <parameterList> ) ; ) /* constructor */

```

TYPES

```

<typeRef> := <baseType> | <namedType> | <tupleType>

<namedType> := <typeName> | <specializedType> | <headerStackType>

<prefixedType> := { . } TYPE_IDENTIFIER

<tupleType> := tuple < <typeArgumentList> >

<headerStackType> := <typeName> [ <expression> ]

<specializedType> := <typeName> < <typeArgumentList> >

<baseType> :=
    bool | error | string | void |
    int { < <integerTypeSize> > } |
    bit { < <integerTypeSize> > } |
    varbit < <integerTypeSize> >

<integerTypeSize> := INTEGER /* | '(' <expression> ')' */

<typeOrVoid> := <typeRef> | void | IDENTIFIER /* type variable */

<optTypeParameters> := { < <typeParameterList> > }

```

```

<typeParameterList> := <name> { , <name> }*

<realTypeArg> := _ | <typeRef>

<typeArg> := _ | <typeRef> | <nonTypeName>

<realTypeArgumentList> := <realTypeArg> { , <realTypeArg> }*

<typeArgumentList> := { <typeArg> { , <typeArg> }* }

<typeDeclaration> := <derivedTypeDeclaration> | <typedefDeclaration> | ( <parserTypeDeclaration> ; )
    | ( <controlTypeDeclaration> ; ) | ( <packageTypeDeclaration> ; )

<derivedTypeDeclaration> := <headerTypeDeclaration> | <headerUnionDeclaration> |
    <structTypeDeclaration> | <enumDeclaration>

<headerTypeDeclaration> := header <name> { <structFieldList> }

<headerUnionDeclaration> := header_union <name> { <structFieldList> }

<structTypeDeclaration> := struct <name> { <structFieldList> }

<structFieldList> := { <structField> { , <structField> }* }

<structField> := <typeRef> <name> ;

<enumDeclaration> := enum { bit < INTEGER > } <name> { <specifiedIdentifierList> }

<errorDeclaration> := error { <identifierList> }

<matchKindDeclaration> := match_kind { <identifierList> }

<identifierList> := <name> { , <name> }*

<specifiedIdentifierList> := <specifiedIdentifier> { , <specifiedIdentifier> }*

<specifiedIdentifier> := <name> { = <initializer> }

<typedefDeclaration> := ( typedef | type ) ( <typeRef> | <derivedTypeDeclaration> ) <name> ';'

```

STATEMENTS

```

<assignmentOrMethodCallStatement> := <lvalue>
    ( { < <typeArgumentList> > } ( <argumentList> ) ; ) |
    ( = <expression> ; )

<emptyStatement> := ;

<returnStatement> := return { expression } ;

<exitStatement> := exit ;

<conditionalStatement> := if ( <expression> ) <statement> { else <statement> }

/* To support direct invocation of a control or parser without instantiation. */
<directApplication> := <typeName> . apply ( <argumentList> ) ;

<statement> := <assignmentOrMethodCallStatement> | <directApplication> | <conditionalStatement> |
    <emptyStatement> | <blockStatement> | <exitStatement> | <returnStatement> | <switchStatement>

<blockStatement> := { <statementOrDeclList> }

```

```

<statementOrDeclList> := { <statementOrDeclaration> }*

<switchStatement> := switch ( <expression> ) { <switchCases> }

<switchCases> := { <switchCase> }*

<switchCase> := <switchLabel> : { <blockStatement> }

<switchLabel> := <name> | default

<statementOrDeclaration> := <variableDeclaration> | <statement> | <instantiation>

```

TABLES

```

<tableDeclaration> := table <name> { <tablePropertyList> }

<tablePropertyList> := <tableProperty> { <tableProperty> }*

<tableProperty> :=
  ( key = { <keyElementList> } ) |
  ( actions = { <actionList> } ) |
  ( const entries = { <entriesList> } ) | /* immutable entries */
  ( { const } <nonTableKwName> = <initializer> ; )

<keyElementList> := { <keyElement> }*

<keyElement> := <expression> : <name> ;

<actionList> := { <actionRef> ; }*

<actionRef> := <prefixedNonTypeName> { ( <argumentList> ) }

<entriesList> := <entry> { <entry> }*

<entry> := <keysetExpression> : <actionRef> ;

<actionDeclaration> := action <name> ( <parameterList> ) <blockStatement>

```

VARIABLES

```

<variableDeclaration> := { const } <typeRef> <name> { = <expression> };

```

EXPRESSIONS

```

<functionDeclaration> := <functionPrototype> <blockStatement>

<argumentList> := { <argument> { , <argument> }* }

<argument> := <expression> | _

/* <kvList> := <kvPair> { ',' <kvPair> }* */

/* <kvPair> := <name> = <expression> */

<expressionList> := { <expression> { , <expression> }* }

<prefixedNonTypeName> := { . } <nonTypeName>

<lvalue> := <prefixedNonTypeName> {
  ( . <name> ) | /* member selector */
  ( [ <indexExpression> ] ) /* array subscript */
}*

```

```

<expression> := <expressionPrimary> { <exprOperator> <expression> }*

<expressionPrimary> := <integer> | <boolean> | <string> |
  ( { . } <nonTypeName> ) |
  ( { <expressionList> } ) | /* <kvList> */
  ( ( <expression> ) ) |
  ( ( ! | ~ | - ) <expression> ) | /* unary expression */
  ( <namedType> | error ) | /* member selector, function call */
  ( ( <typeRef> ) <expression> ) /* cast */

<exprOperator> := <binaryOperator> |
  ( . <name> ) | /* member selector */
  ( [ <indexExpression> ] ) | /* array subscript */
  ( ( <argumentList> ) ) | /* function call */
  ( < <realTypeArgumentList> > ) |
  ( = <expression> ) /* <kvPair> */

<indexExpression> := <expression> { : <expression> }

<integer> := INTEGER

<boolean> := true | false

<string> := STRING

<binaryOperator> := * | / | + | - | <= | >= | < | > | != | == | || | && | | | & | << | >>

```

Syntax Tree

PROGRAM

1. `<p4program> := <declarationList>decl_list`
2. `<declarationList> := { <declaration>[0..n] }*`
3. `<declaration> := (<variableDeclaration> | <externDeclaration> | <actionDeclaration> | <functionDeclaration> | <parserDeclaration> | <parserTypeDeclaration> | <controlDeclaration> | <controlTypeDeclaration> | <typeDeclaration> | <errorDeclaration> | <matchKindDeclaration> | <instantiation>)decl`
4. `<name> := STRINGstrname`
5. `<parameterList> := { <parameter>[0..n] }*`
6. `<parameter> := { in | out | inout }direction <typeRef>type <name>name { <expression> }init_expr`
7. `<packageTypeDeclaration> := <name>name { <typeParameterList> }type_params <parameterList>params`
8. `<instantiation> := <typeRef>type_ref <argumentList>args <name>name`

PARSER

1. `<parserDeclaration> := <typeDeclaration>proto { <parameterList> }ctor_params`
`<parserLocalElements>local_elements <parserStates>states`
2. `<parserTypeDeclaration> := <name>name { <typeParameterList> }type_params <parameterList>params`
3. `<parserLocalElements> := { <parserLocalElement>[0..n] }*`

4. `<parserLocalElement> := (<variableDeclaration> | <instantiation>)element`
5. `<parserStates> := { <parserState>[0..n] }+`
6. `<parserState> := <name>name <parserStatements>stmt_list <transitionStatement>transition_stmt`
7. `<parserStatements> := { <parserStatement>[0..n] }*`
8. `<parserStatement> := (<assignmentStatement> | <functionCall> | <directApplication> | <parserBlockStatement> | <variableDeclaration>)stmt`
9. `<parserBlockStatement> := <parserStatements>stmt_list`
10. `<transitionStatement> := <stateExpression>stmt`
11. `<stateExpression> := (<name> | <selectExpression>)expr`
12. `<selectExpression> := <expressionList>expr_list <selectCaseList>case_list`
13. `<selectCaseList> := { <selectCase>[0..n] }*`
14. `<selectCase> := <keysetExpression>keyset_expr <name>name`
15. `<keysetExpression> := (<tupleKeysetExpression> | <expression> | <defaultKeysetExpression> | <dontcareKeysetExpression>)expr`
16. `<tupleKeysetExpression> := <keysetExpressionList>expr_list`
17. `<keysetExpressionList> := { <simpleKeysetExpression>[0..n] }+`

CONTROL

1. `<controlDeclaration> := <typeDeclaration>proto { <parameterList> }ctor_params <controlLocalDeclarations>local_decls <blockStatement>apply_stmt`
2. `<controlTypeDeclaration> := <name>name { <typeParameterList> }type_params <parameterList>params`
3. `<controlLocalDeclarations> := { <controlLocalDeclaration>[0..n] }*`
4. `<controlLocalDeclaration> := (<variableDeclaration> | <actionDeclaration> | <tableDeclaration> | <instantiation>)decl`

EXTERN

1. `<externDeclaration> := (<externTypeDeclaration> | <functionPrototype>)decl`
2. `<externTypeDeclaration> := <name>name { <typeParameterList>type_params } <methodPrototypes>method_protos`
3. `<methodPrototypes> := { <methodPrototype>[0..n] }*`
4. `<functionPrototype> := { <typeRef> }return_type <name>name { <typeParameterList> }type_params <parameterList>params`

TYPES

1. `<typeRef> := (<baseTypeBool> | <baseTypeError> | <baseTypeInteger> | <baseTypeBit> | <baseTypeVarbit> | <baseTypeString> | <baseTypeVoid> | <namedType> | <tupleType>)type`
2. `<namedType> := (<name> | <specializedType> | <headerStackType>)type`

3. <tupleType> := <typeArgumentList>_{type_args}
4. <headerStackType> := <name>_{name} <expression>_{stack_expr}
5. <specializedType> := <name>_{name} <typeArgumentList>_{type_args}
6. <baseTypeBool> := <name>_{name}
7. <baseTypeInteger> := <name>_{name} { <integerTypeSize> }_{size}
8. <baseTypeBit> := <name>_{name} { <integerTypeSize> }_{size}
9. <baseTypeVarbit> := <name>_{name} <integerTypeSize>_{size}
10. <baseTypeString> := <name>_{name}
11. <baseTypeVoid> := <name>_{name}
12. <baseTypeError> := <name>_{name}
13. <integerTypeSize> := INTEGER_{size}
14. <typeParameterList> := { <name>[_0..n] }₊
15. <realTypeArg> := (_ | <typeRef>)_{arg}
16. <typeArg> := (_ | <typeRef> | <name>)_{arg}
17. <realTypeArgumentList> := { <realTypeArg>[_0..n] }₊
18. <typeArgumentList> := { <typeArg>[_0..n] }_{*}
19. <typeDeclaration> := (<derivedTypeDeclaration> | <typedefDeclaration> | <parserTypeDeclaration> | <controlTypeDeclaration> | <packageTypeDeclaration>)_{decl}
20. <derivedTypeDeclaration> := (<headerTypeDeclaration> | <headerUnionDeclaration> | <structTypeDeclaration> | <enumDeclaration>)_{decl}
21. <headerTypeDeclaration> := <name>_{name} <structFieldList>_{fields}
22. <headerUnionDeclaration> := <name>_{name} <structFieldList>_{fields}
23. <structTypeDeclaration> := <name>_{name} <structFieldList>_{fields}
24. <structFieldList> := { <structField>[_0..n] }_{*}
25. <structField> := <typeRef>_{type} <name>_{name}
26. <enumDeclaration> := INTEGER_{type_size} <name>_{name} <specifiedIdentifierList>_{fields}
27. <errorDeclaration> := <identifierList>_{fields}
28. <matchKindDeclaration> := <identifierList>_{fields}
29. <identifierList> := { <name>[_0..n] }₊
30. <specifiedIdentifierList> := { <specifiedIdentifier>[_0..n] }₊
31. <specifiedIdentifier> := <name>_{name} { <initializer> }_{init_expr}
32. <typedefDeclaration> := (<typeRef> | <derivedTypeDeclaration>)_{type_ref} <name>_{name}

STATEMENTS

1. `<assignmentStatement> := <lvalueExpression>lhs_expr <expression>rhs_expr`
2. `<functionCall> := (<expression> | <lvalueExpression>)lhs_expr <argumentList>args`
3. `<returnStatement> := { <expression> }expr`
4. `<exitStatement>`
5. `<conditionalStatement> := <expression>cond_expr <statement>stmt <statement>else_stmt`
6. `<directApplication> := <typeName>name <argumentList>args`
7. `<statement> := (<assignmentStatement> | <functionCall> | <directApplication> | <conditionalStatement> | <emptyStatement> | <blockStatement> | <exitStatement> | <returnStatement> | <switchStatement>)stmt`
8. `<blockStatement> := <statementOrDeclList>stmt_list`
9. `<statementOrDeclList> := { <statementOrDeclaration>[0..n] }*`
10. `<switchStatement> := <expression>expr <switchCases>switch_cases`
11. `<switchCases> := { <switchCase>[0..n] }*`
12. `<switchCase> := <switchLabel>label <blockStatement>stmt`
13. `<switchLabel> := (<name> | default)label`
14. `<statementOrDeclaration> := (<variableDeclaration> | <statement> | <instantiation>)stmt_or_decl`

TABLES

1. `<tableDeclaration> := <name>name <tablePropertyList>prop_list`
2. `<tablePropertyList> := { tableProperty[0..n] }+`
3. `<tableProperty> := (<keyProperty> | <actionsProperty> | <entriesProperty> | <simpleProperty>)prop`
4. `<keyProperty> := <keyElementList>keyelem_list`
5. `<keyElementList> := { <keyElement>[0..n] }*`
6. `<keyElement> := <expression>expr <name>match`
7. `<actionsProperty> := <actionsList>action_list`
8. `<actionList> := { <actionRef>[0..n] }*`
9. `<actionRef> := <name>name <argumentList>args`
10. `<entriesProperty> := <entriesList>entries_list`
11. `<entriesList> := { <entry>[0..n] }+`
12. `<entry> := <keysetExpression>keyset <actionRef>action`

13. `<simplePropertey> := <name>name <expression>init_expr`

14. `<actionDeclaration> := <name>name <parameterList>params <blockStatement>stmt`

VARIABLES

1. `<variableDeclaration> := <typeRef>type <name>name { <expression> }init_expr`

EXPRESSIONS

1. `<functionDeclaration> := <functionPrototype>proto <blockStatement>stmt`

2. `<argumentList> := { <argument>[_0..n] }*`

3. `<argument> := (<expression> | -)arg`

4. `<kvPair> := <name>name <expression>init_expr`

5. `<expressionList> := { <expression>[_0..n] }*`

6. `<lvalueExpression> := (<name> | <memberSelector> | <arraySubscript>)expr`

7. `<expression> := (<integerLiteral> | <booleanLiteral> | <stringLiteral> | <name> | <namedType> | <expressionList> | castExpression | <unaryExpression> | <binaryExpression> | <memberSelector> | <arraySubscript> | <functionCall> | <kvPair>)expr { <realTypeArgumentList> }type_args`

8. `<castExpression> := <typeRef>type <expression>expr`

9. `<unaryExpression> := OPERATORop <expression>operand`

10. `<binaryExpression> := <expression>left_operand OPERATORop <expression>right_operand`

11. `<memberSelector> := <expression>lhs_expr <name>name`

12. `<arraySubscript> := <expression>lhs_expr <indexExpression>index_expr`

13. `<indexExpression> := <expression>start_index { <expression> }end_index`

14. `<integerLiteral> := INTEGERvalue INTEGERwidth`

15. `<booleanLiteral> := INTEGERvalue`

16. `<stringLiteral> := STRINGvalue`